
1�� Getting Your Feet Wet
With mod_perl 2.0 

1.1��About
Prerequisites

Installation

Configuration

Server Launch and Shutdown

Registry Scripts

Handler Modules

1.2��Prerequisites
Apache 2.0 is required.

mod_perl DSO: 2.0.47

static mod_perl: 2.0.51

Perl, depending on MPM:

prefork: 5.6.1 (no ithreads required)

threaded: 5.8.0 w/ithreads

The installation details are in the handouts

Static prefork build

5.6.1 or higher w/ ithreads

5.8.2 or higher w/o ithreads

Static threaded build

5.8.0 or higher w/ ithreads

threads.pm

Perl 5.8.1 or higher w/ithreads enabled

1.3��mod_perl Installation
  % lwp-download \
    http://perl.apache.org/dist/mod_perl-2.0.01.tar.gz
  % tar -xvzf mod_perl-2.x.xx.tar.gz
  % cd modperl-2.0
  % perl Makefile.PL MP_APXS=$HOME/httpd/bin/apxs
  % make && make test && make install

MP_APXS is a full path to the apxs  executable

1.4��Configuration
Enable DSO:

  LoadModule perl_module modules/mod_perl.so

Enable the 1.0 compatibility layer

  PerlModule Apache2::compat

1.5��Server Launch and 
Shutdown

Start:

  % $HOME/httpd/prefork/bin/apachectl start

  % tail -f $HOME/httpd/prefork/logs/error_log

  [Fri Jul 22 09:39:55 2005] [notice] Apache/2.0.55-dev (Unix)
  mod_ssl/2.0.55-dev OpenSSL/0.9.7e DAV/2 mod_perl/2.0.2-dev
  Perl/v5.8.7 configured -- resuming normal operations

Stop:

  % $HOME/httpd/prefork/bin/apachectl stop

1.6��Registry Scripts
Configuration

  Alias /perl/ /home/httpd/httpd-2.0/perl/
  <Location /perl/>
      SetHandler perl-script
      PerlResponseHandler ModPerl::Registry
      PerlOptions +ParseHeaders
      Options +ExecCGI
  </Location>

Restart the server

A simple script:

  #!/usr/bin/perl
  print "Content-type: text/plain\n\n";
  print "mod_perl 2.0 rocks!\n";

Make it executable and readable:

  % chmod a+rx /home/httpd/httpd-2.0/perl/rock.pl

Request:

  % lwp-request http://localhost/perl/rock.pl
  mod_perl 2.0 rocks!

1.7��Handler Modules
  # /home/httpd/httpd-2.0/perl/MyApache2/Rocks.pm                      #
  package MyApache2::Rocks;
  
  use strict;
  use warnings;
  
  use Apache2::RequestRec ();
  use Apache2::RequestIO ();
  
  use Apache2::Const -compile => qw(OK);
  
  sub handler {
      my $r = shift;
  
      $r->content_type('text/plain');
      print "mod_perl 2.0 rocks!\n";
  
      return Apache2::Const::OK;
  }
  1;

Adjust @INC via a startup file

  use lib qw(/home/httpd/httpd-2.0/perl);

loaded from httpd.conf:

  PerlRequire /home/httpd/httpd-2.0/perl/startup.pl

Configure:

  <Location /rocks>
      SetHandler perl-script
      PerlResponseHandler  MyApache2::Rocks
  </Location>

Restart server

Issue request:

  % lwp-request http://localhost/rocks
  mod_perl 2.0 rocks!

2�� New Concepts 

2.1��About
Exceptions

Bucket Brigades

2.2��Exceptions
Apache and APR API return a status code for almost all 
methods

In C you must check return value of each call

It makes it really hard/slow to prototype things

It makes the code hard to read

It's easy to forget to check the return values -- causing
problems later

It complicates the API, when you may want to return other 
values

APR::Error  handles APR/Apache/mod_perl exceptions for
you, while leaving you in control.

Instead of returning status codes, we throw exceptions

If you don't catch those exceptions, everything works 
transparently

  my $rlen = $sock->recv(my $buff, 1024);

Perl will intercept the exception object and die()  with a
proper error message.

  [Wed Jun 09 07:25:25 2004] [error] [client 127.0.0.1]
  APR::Socket::recv: (11) Resource temporarily unavailable
  at .../TestError/runtime.pm line 124

If you want to catch an exception -- you can.

  use APR::Status ();                                              #
  # ....
  my $tries = 0;
  RETRY: my $rlen = eval { $sock->recv(my $buffer, SIZE) };
  if ($@) {
      die $@ unless ref $@ && APR::Status::is_EAGAIN($@);
      if ($tries++ < 3) {
          # sleep 250msec
          select undef, undef, undef, 0.25;
          goto RETRY;
      }
      else {
          # do something else
      }
  }
  warn "read $rlen bytes\n"

APR::Status::is_FOO($@) is for complicated cases

most of the time it'll be a simple compare:

  use APR::Const -compile => qw(TIMEUP);
  # some code throwing an exception
  if ($@ && ref $@ && $@ == APR::Const::TIMEUP) {
      # do something
  }

APR::Error  uses Perl operator overloading:

in boolean and numerical contexts -- gives the status 
code

in the string context -- the full error message

it is called APR::Error  as it is used by

mod_perl

APR apps written in Perl

2.3��Bucket Brigades

Apache 2.0 implements request and response data flow 
filtering

Modules can filter each other's output

No need to modify Apache to accomodate SSL,
compressions, transformation filters.

The Bucket Brigades technology was introduced to make IO
filtering efficient and avoid unnecessary copying.

Buckets :

A bucket represents a chunk of data.

Buckets linked together comprise a brigade.

Each bucket in a brigade can be modified, removed and
replaced with another bucket.

Bucket types: files, data blocks, flush and end of stream
indicators, pools, etc.

To manipulate a bucket one doesn't need to know its internal 
representation.

Bucket Brigades :

A stream of data is represented by bucket brigades. 

Filters manipulate brigades one by one
(adding/modifying/removing buckets)

... and pass the brigade to the next filter on the stack

Here's an imaginary bucket brigade after it has passed
through several filters.

More about BBs when we talk about protocols and filters

3�� Introducing mod_perl
Handlers 

3.1��About
Handler Anatomy

mod_perl Handlers Categories

Single Phase's Multiple Handlers Behavior

3.2��Handler Anatomy
Apache distinguishes between numerous phases

Each phase provides a hook: ap_hook_<phase_name>

These hooks are used by modules to alter the default Apache 
behavior

Hooks are usually referred to as handlers or callbacks

Naming convention: PerlFooHandler

e.g. PerlResponseHandler  configures the response 
callback.

  package MyApache2::CurrentTime;
  
  use strict;
  use warnings;
  
  use Apache2::RequestRec ();
  use Apache2::RequestIO ();
  
  use Apache2::Const -compile => qw(OK);
  
  sub handler {
      my $r = shift;
  
      $r->content_type('text/plain');
      $r->print("The time is: " . scalar(localtime) . "\n");
  
      return Apache2::Const::OK;
  }
  1;

Configuring the response handler:

  PerlModule MyApache2::CurrentTime
  <Location /time>
      SetHandler modperl
      PerlResponseHandler MyApache2::CurrentTime
  </Location>

A request to http://localhost/time returns the curent time

Stas BekmanIntroducing mod_perl Handlers Slide 39

http://localhost/time


3.3��mod_perl Handler 
Categories
The mod_perl handlers can be divided by their application scope
in several categories:

Server life cycle  
PerlOpenLogsHandler  
PerlPostConfigHandler  
PerlChildInitHandler  
PerlChildExitHandler

Protocols  
PerlPreConnectionHandler  
PerlProcessConnectionHandler

Filters  
PerlInputFilterHandler  
PerlOutputFilterHandler

HTTP Protocol  
PerlPostReadRequestHandler  
PerlTransHandler  
PerlMapToStorageHandler  
PerlInitHandler  
PerlHeaderParserHandler  
PerlAccessHandler  
PerlAuthenHandler  
PerlAuthzHandler  
PerlTypeHandler  
PerlFixupHandler  
PerlResponseHandler  
PerlLogHandler  
PerlCleanupHandler

3.4��Stacked Handlers
Phases' behavior varies when there is more then one handler
registered to run for the same phase.

The following table specifies each handler's behavior in this 
situation:

    Directive                   Type                                      #
  --------------------------------------
  PerlOpenLogsHandler          RUN_ALL
  PerlPostConfigHandler        RUN_ALL
  PerlChildInitHandler         VOID
  PerlChildExitHandler         VOID
  
  PerlPreConnectionHandler     RUN_ALL
  PerlProcessConnectionHandler RUN_FIRST
  
  PerlPostReadRequestHandler   RUN_ALL
  PerlTransHandler             RUN_FIRST
  PerlMapToStorageHandler      RUN_FIRST
  PerlInitHandler              RUN_ALL
  PerlHeaderParserHandler      RUN_ALL
  PerlAccessHandler            RUN_ALL
  PerlAuthenHandler            RUN_FIRST
  PerlAuthzHandler             RUN_FIRST
  PerlTypeHandler              RUN_FIRST
  PerlFixupHandler             RUN_ALL
  PerlResponseHandler          RUN_FIRST
  PerlLogHandler               RUN_ALL
  PerlCleanupHandler           RUN_ALL
  
  PerlInputFilterHandler       VOID
  PerlOutputFilterHandler      VOID

The types :

VOID

Executed in the order they have been registered
disregarding their return values.

Though in mod_perl they are expected to return 
Apache2::Const::OK .

RUN_ALL

Executed in the order they have been registered until the
first handler that returns something other than 
Apache2::Const::OK  or 
Apache2::Const::DECLINED .

RUN_FIRST

Executed in the order they have been registered until the
first handler that returns something other than 
Apache2::Const::DECLINED .

If the return value is Apache2::Const::DECLINED , the
next handler in the chain will be run.

If the return value is Apache2::Const::OK  the next
phase will start.

In all other cases the execution will be aborted.

4�� Server Life Cycle
Handlers 

4.1��About
Server Life Cycle

Startup Phases Demonstration Module

PerlOpenLogsHandler

PerlPostConfigHandler

PerlChildInitHandler

PerlChildExitHandler

4.2��Apache 2.0 Server Life 
Cycle

1.  Parse the configuration file, open_logs, post_config

2.  Restart to test graceful restarts

3.  Parse the configuration file, open_logs, post_config

4.  Spawn workers: procs, threads, mix

Run child_init for each spawned process

5.  Process requests over connections

6.  Shutdown: child_exit

4.2.1��Startup Phases
Demonstration Module
  package MyApache2::StartupLog;                                    #
  
  use strict;
  use warnings;
  
  use Apache2::Log ();
  use Apache2::ServerUtil ();
  
  use Fcntl qw(:flock);
  use File::Spec::Functions;
  
  use Apache2::Const -compile => 'OK';
  
  my $log_path = catfile Apache2::ServerUtil::server_root,
      "logs", "startup_log";
  my $log_fh;

  sub open_logs {
      my($conf_pool, $log_pool, $temp_pool, $s) = @_;
  
      $s->warn("opening the log file: $log_path");
      open $log_fh, ">>$log_path" or die "can't open $log_path: $!";
      my $oldfh = select($log_fh); $| = 1; select($oldfh);
  
      say("process $$ is born to reproduce");
      return Apache2::Const::OK;
  }
  
  sub post_config {
      my($conf_pool, $log_pool, $temp_pool, $s) = @_;
      say("configuration is completed");
      return Apache2::Const::OK;
  }
  
  sub child_init {                                                #
      my($child_pool, $s) = @_;
      say("process $$ is born to serve");
      return Apache2::Const::OK;
  }

  sub child_exit {                                                #
      my($child_pool, $s) = @_;
      say("process $$ now exits");
      return Apache2::Const::OK;
  }
  
  sub say {
      my($caller) = (caller(1))[3] =~ /([^:]+)$/;
      if (defined $log_fh) {
          flock $log_fh, LOCK_EX;
          printf $log_fh "[%s] - %-11s: %s\n", 
              scalar(localtime), $caller, $_[0];
          flock $log_fh, LOCK_UN;
      }
      else {
          # when the log file is not open
          warn __PACKAGE__ . " says: $_[0]\n";
      }
  }

  my $parent_pid = $$;                                            #
  END {
      my $msg = "process $$ is shutdown";
      $msg .= "\n". "-" x 20 if $$ == $parent_pid;
      say($msg);
  }
  
  1;

And the httpd.conf configuration section:

  PerlModule            MyApache2::StartupLog
  PerlOpenLogsHandler   MyApache2::StartupLog::open_logs
  PerlPostConfigHandler MyApache2::StartupLog::post_config
  PerlChildInitHandler  MyApache2::StartupLog::child_init
  PerlChildExitHandler  MyApache2::StartupLog::child_exit

  % bin/apachectl start

  # logs/startup_log>:
  [Sun Jun  6 01:50:06 2004] - open_logs  : process 24189 is born to reproduce
  [Sun Jun  6 01:50:06 2004] - post_config: configuration is completed
  [Sun Jun  6 01:50:07 2004] - END        : process 24189 is shutdown
  --------------------
  [Sun Jun  6 01:50:08 2004] - open_logs  : process 24190 is born to reproduce
  [Sun Jun  6 01:50:08 2004] - post_config: configuration is completed
  [Sun Jun  6 01:50:09 2004] - child_init : process 24192 is born to serve
  [Sun Jun  6 01:50:09 2004] - child_init : process 24193 is born to serve
  [Sun Jun  6 01:50:09 2004] - child_init : process 24194 is born to serve
  [Sun Jun  6 01:50:09 2004] - child_init : process 24195 is born to serve

Apache restarts itself

  % bin/apachectl stop

  [Sun Jun  6 01:50:10 2004] - child_exit : process 24193 now exits
  [Sun Jun  6 01:50:10 2004] - END        : process 24193 is shutdown
  [Sun Jun  6 01:50:10 2004] - child_exit : process 24194 now exits
  [Sun Jun  6 01:50:10 2004] - END        : process 24194 is shutdown
  [Sun Jun  6 01:50:10 2004] - child_exit : process 24195 now exits
  [Sun Jun  6 01:50:10 2004] - child_exit : process 24192 now exits
  [Sun Jun  6 01:50:10 2004] - END        : process 24192 is shutdown
  [Sun Jun  6 01:50:10 2004] - END        : process 24195 is shutdown
  [Sun Jun  6 01:50:10 2004] - END        : process 24190 is shutdown
  --------------------

All processes run the END block on shutdown

The presented behavior varies from MPM to MPM.

I used prefork MPM for this demo

MPMs, like winnt, may run open_logs and post_config more
than once

the END blocks may be run more times, when threads are 
involved.

The only sure thing -- each of these phases runs at least once

5�� Protocol Handlers 

5.1��About
Connection Cycle Phases

PerlPreConnectionHandler

PerlProcessConnectionHandler

Socket-based Protocol Module

Bucket Brigades-based Protocol Module

Stas BekmanProtocol Handlers Slide 62



5.2��Connection Cycle Phases
Each child server is engaged in processing connections.

Each connection may be served by different connection 
protocols,

e.g., HTTP, POP3, SMTP, etc.

Each connection may include more than one request,

e.g., several HTTP requests can be served over a single
connection, when a response includes several images.

Stas BekmanProtocol Handlers Slide 63



Connection Life Cycle  diagram:

5.2.1��PerlPreConnectionHandler
 

The pre_connection phase happens just after the server
accepts the connection, but before it is handed off to a
protocol module to be served.

It gives modules an opportunity to modify the connection as
soon as possible and insert filters if needed.

The core server uses this phase to setup the connection
record based on the type of connection that is being used.

mod_perl itself uses this phase to register the connection
input and output filters.

Apache2::Reload :

In mod_perl 1.0 Apache::Reload  was used to automatically
reload modified since the last request Perl modules.

It was invoked during post_read_request, the first HTTP
request's phase.

In mod_perl 2.0 pre_connection is the earliest general phase

So now we can invoke the Apache2::Reload  handler
during the pre_connection phase if the interpreter's scope is:

  PerlInterpScope connection

Though this is not good for a production server,

where there are several requests coming on the same 
connection

and only one handled by mod_perl

and the others by the default images handler

the Perl interpreter won't be available to other threads while
the images are being served.

Skeleton :

  sub handler {
      my $c = shift;
      # ...
      return Apache2::Const::OK;
  }

A pre_connection handler accepts one argument:

connection record

Block by IP example :

  package MyApache2::BlockIP2;                                       #
  
  use Apache2::Connection ();
  
  use Apache2::Const -compile => qw(FORBIDDEN OK);
  
  my %bad_ips = map {$_ => 1} qw(127.0.0.1 10.0.0.4);
  
  sub handler {
      my $c = shift;
  
      my $ip = $c->remote_ip;
      if (exists $bad_ips{$ip}) {
          warn "IP $ip is blocked\n";
          return Apache2::Const::FORBIDDEN;
      }
  
      return Apache2::Const::OK;
  }
  1;

Configuration

  PerlPreConnectionHandler MyApache2::BlockIP2

Apache simply drops the connection if the IP is blacklisted

Almost no resources are wasted

5.2.2��PerlProcessConnectionHandler
 

The process_connection phase is used to process incoming 
connections.

Only protocol modules should assign handlers for this phase,
as it gives them an opportunity to replace the standard HTTP
processing with processing for some other protocols (e.g., POP3,
FTP, etc.).

The only way to run protocol servers different than the core
HTTP is inside dedicated virtual hosts.

process_connection skeleton :

  sub handler {
      my ($c) = @_;
      my $sock = $c->client_socket;
      $sock->opt_set(APR::Const::SO_NONBLOCK, 0);
      # ...
      return Apache2::Const::OK;
  }

arg1: a connection record object

a socket object: retrieved from $c

must set the socket to blocking IO mode

Let's look at the following two examples of connection 
handlers:

1.  Using the connection socket to read and write the data.

2.  Using bucket brigades to accomplish the same and allow
for connection filters to do their work.

5.2.2.1��Socket-based Protocol Module

The MyApache2::EchoSocket  module simply echoes the
data read back to the client.

This module's implementation works directly with the
connection socket and therefore bypasses connection filters if 
any.

Configuration :

  Listen 8010
  <VirtualHost _default_:8010>
      PerlModule                   MyApache2::EchoSocket
      PerlProcessConnectionHandler MyApache2::EchoSocket
  </VirtualHost>

use the Listen  and <VirtualHost>  directives to bind to
the non-standard port 8010:

Demo:

  panic% httpd
  panic% telnet localhost 8010
  Trying 127.0.0.1...
  Connected to localhost (127.0.0.1).
  Escape character is '^]'.
  Hello
  Hello
  
  fOo BaR
  fOo BaR

  Connection closed by foreign host.

The code :

  package MyApache2::EchoSocket;                                     #
  
  use strict;
  use warnings FATAL => 'all';
  
  use Apache2::Connection ();
  use APR::Socket ();
  
  use Apache2::Const -compile => 'OK';
  use APR::Const     -compile => 'SO_NONBLOCK';
  
  use constant BUFF_LEN => 1024;

  sub handler {                                                     #
      my $c = shift;
      my $sock = $c->client_socket;
  
      $sock->opt_set(APR::Const::SO_NONBLOCK => 0);
  
      while ($sock->recv(my $buff, BUFF_LEN)) {
          last if $buff =~ /^[\r\n]+$/;
          $sock->send($buff);
      }
  
      Apache2::Const::OK;
  }
  1;

5.2.2.2��Bucket Brigades-based Protocol 
Module

The same module, but this time implemented by manipulating
bucket brigades,

and which runs its output through a connection output filter
that turns all uppercase characters into their lowercase 
equivalents.

Configuration :

  Listen 8011
  <VirtualHost _default_:8011>
      PerlModule                   MyApache2::EchoBB
      PerlProcessConnectionHandler MyApache2::EchoBB
      PerlOutputFilterHandler      MyApache2::EchoBB::lowercase_filter
  </VirtualHost>

Demo:

  panic% httpd
  panic% telnet localhost 8011
  Trying 127.0.0.1...
  Connected to localhost (127.0.0.1).
  Escape character is '^]'.
  Hello
  hello
  
  fOo BaR
  foo bar

As you can see the response now was all in lower case,
because of the output filter.

The code :

  package MyApache2::EchoBB;                                         #
  
  use strict;
  use warnings FATAL => 'all';
  
  use Apache2::Connection ();
  use APR::Socket ();
  use APR::Bucket ();
  use APR::Brigade ();
  use APR::Error ();
  use APR::Status ();
  
  use APR::Const     -compile => qw(SUCCESS SO_NONBLOCK);
  use Apache2::Const -compile => qw(OK MODE_GETLINE);

  sub handler {                                                     #
      my $c = shift;
  
      $c->client_socket->opt_set(APR::Const::SO_NONBLOCK => 0);
  
      my $bb_in  = APR::Brigade->new($c->pool, $c->bucket_alloc);
      my $bb_out = APR::Brigade->new($c->pool, $c->bucket_alloc);
  
      my $last = 0;
      while (1) {
          my $rc = $c->input_filters->get_brigade($bb_in,
                                                  Apache2::Const::MODE_GETLINE);
          last if APR::Status::is_EOF($rc);
          die APR::Error::strerror($rc) unless $rc == APR::Const::SUCCESS;

          while (!$bb_in->empty) {                                  #
              my $b = $bb_in->first;
  
              $b->remove;
  
              if ($b->is_eos) {
                  $bb_out->insert_tail($b);
                  last;
              }
  
              if ($b->read(my $data)) {
                  $last++ if $data =~ /^[\r\n]+$/;
                  # could do some transformation on data here
                  $b = APR::Bucket->new($bb_out->bucket_alloc, $data);
              }
  
              $bb_out->insert_tail($b);
          }

          my $fb = APR::Bucket::flush_create($c->bucket_alloc);     #
          $bb_out->insert_tail($fb);
          $c->output_filters->pass_brigade($bb_out);
          last if $last;
      }
  
      $bb_in->destroy;
      $bb_out->destroy;
  
      Apache2::Const::OK;
  }

  use base qw(Apache2::Filter);                                      #
  use constant BUFF_LEN => 1024;
  
  sub lowercase_filter : FilterConnectionHandler {
      my $filter = shift;
    
      while ($filter->read(my $buffer, BUFF_LEN)) {
          $filter->print(lc $buffer);
      }
    
      return Apache2::Const::OK;
  }
  
  1;

A simplified pseudo-code handler:

  while ($bb_in = get_brigade()) {
  
      while ($b_in = $bb_in->get_bucket()) {                   #
          $b_in->read(my $data);
  
          # do something with $data
  
          $b_out = new_bucket($data);
  
          $bb_out->insert_tail($b_out);
      }
  
      $bb_out->insert_tail($flush_bucket);
      pass_brigade($bb_out);
  }

Use fflush  -- replace:

          my $fb = APR::Bucket::flush_create($c->bucket_alloc);     #
          $bb_out->insert_tail($fb);
          $c->output_filters->pass_brigade($bb_out);

with just one line:

          $c->output_filters->fflush($bb_out);                     #

The same handler using a single bucket brigade

  sub handler {
      my $c = shift;
      $c->client_socket->opt_set(APR::Const::SO_NONBLOCK, 0);
      my $bb = APR::Brigade->new($c->pool, $c->bucket_alloc);
  
      while (1) {
          my $rc = $c->input_filters->get_brigade($bb,
                                                  Apache2::Const::MODE_GETLINE);
          last if APR::Status::is_EOF($rc);
          die APR::Error::strerror($rc) unless $rc == APR::Const::SUCCESS;
  
          for (my $b = $bb->first; $b; $b = $bb->next($b)) {
  
              last if $b->is_eos;
  
              if ($b->read(my $data)) {
                  last if $data =~ /^[\r\n]+$/;
                  my $nb = APR::Bucket->new($bb->bucket_alloc, $data);
                  # head->...->$nb->$b ->...->tail
                  $b->insert_before($nb);
                  $b->remove;
              }
          }
  
          $c->output_filters->fflush($bb);
      }

      $bb->destroy;
  
      Apache2::Const::OK;
  }

This handler could be much simpler, since we don't modify
the data.

      $c->client_socket->opt_set(APR::Const::SO_NONBLOCK => 0);
  
      my $bb = APR::Brigade->new($c->pool, $c->bucket_alloc);
      while (1) {
          my $rc = $c->input_filters->get_brigade($bb,
                                                  Apache2::Const::MODE_GETLINE);
          last if APR::Status::is_EOF($rc);
          die APR::Error::strerror($rc) unless $rc == APR::Const::SUCCESS;
  
          $c->output_filters->fflush($bb);
      }
      $bb->destroy;

but it won't work in the telnet mode, since /[\r\n]/  that we
type to end the line will keep this loop running forever

And here is another version which slurps all the data and
works with telnet

      my $bb = APR::Brigade->new($c->pool, $c->bucket_alloc);
      while (1) {
          my $rc = $c->input_filters->get_brigade($bb,
                                                  Apache2::Const::MODE_GETLINE);
          last if APR::Status::is_EOF($rc);
          die APR::Error::strerror($rc) unless $rc == APR::Const::SUCCESS;
  
          next unless $bb->flatten(my $data);
          $bb->cleanup;
          last if $data =~ /^[\r\n]+$/;
          my $b = APR::Bucket->new($bb->bucket_alloc, $data);
          $bb->insert_tail($b);
  
          $c->output_filters->fflush($bb);
      }
      $bb->destroy;

We will discuss the filters next

6�� Input and Output Filters 

6.1��About
Your First Filter

I/O Filtering Concepts

Two Methods for Manipulating Data

HTTP Request Versus Connection Filters

Multiple Invocations of Filter Handlers

Blocking Calls

mod_perl Filters Declaration and Configuration

Filter Priority Types

PerlInputFilterHandler

PerlOutputFilterHandler

PerlSetInputFilter

PerlSetOutputFilter

HTTP Request vs. Connection Filters

Filter Initialization Phase

All-in-One Filter

Input Filters Examples

Output Filters Examples

6.2��Your First Filter
You certainly already know how filters work.

That's because you encounter filters so often in real life.

Why using I/O filters?

Instead of making the response handler code complex

adjust its output with filters:

each doing a simple transformation

and can be:

stacked

repeated

added/removed dynamically at run-time

6.3��A Few Quick Examples
A typical Configuration

  <Location ...>
      PerlOutputFilterHandler Modulename::handlername
  </Location>

The simplest pass through filter

  package MyApache2::FilterTryPassThrough;                      #
  
  use strict;
  use warnings FATAL => 'all';
  
  use Apache2::Const -compile => qw(DECLINED);
  
  sub filter {
      #warn __PACKAGE__ . ": filter called\n";
  
      return Apache2::Const::DECLINED;
  }
  
  1;

Another pass through filter

  package MyApache2::FilterTryPassThrough2;                     #
  
  use strict;
  use warnings FATAL => 'all';
  
  use Apache2::Filter ();
  
  use Apache2::Const -compile => qw(OK);
  
  sub filter {
      my $f = shift;
  
      while ($f->read(my $buffer, 1024)){
          $f->print($buffer);
      }
  
      return Apache2::Const::OK;
  }
  1;

A lowercase filter

convert:

  <HTML><body></body></HTML>

to:

  <html><body></body></html>

  package MyApache2::FilterTryLC;                               #
  
  use strict;
  use warnings FATAL => 'all';
  
  use Apache2::Filter ();
  
  use Apache2::Const -compile => qw(OK);
  
  sub filter {
      my $f = shift;
  
      while ($f->read(my $buffer, 1024)){
          $f->print(lc $buffer);
      }
  
      return Apache2::Const::OK;
  }
  
  1;

A simple web ad blocker filter

turn:

  <html>
    <body>
      <img src="ad.png">
    </body>
  </html>

into:

  <html>
    <body>
  
    </body>
  </html>

  package MyApache2::FilterTryAdBlock;                          #
  
  use strict;
  use warnings FATAL => 'all';
  
  use Apache2::Filter ();
  use Apache2::RequestRec ();
  use APR::Table ();
  
  use Apache2::Const -compile => qw(OK);

  sub filter {                                                  #
      my $f = shift;
  
      unless ($f->ctx) {
          $f->r->headers_out->unset('Content-Length');
          $f->ctx(1);
      }
  
      while ($f->read(my $buffer, 1024)) {
          $buffer =~ s|<img[^>]+>||ig;
          $f->print($buffer);
      }
  
      return Apache2::Const::OK;
  }
  
  1;

A more robust simple web ad blocker filter

deal with "sliced" input

  <html>
    <body>
      <img

followed by:

  src="ad.png">
    </body>
  </html>

  package MyApache2::FilterTryAdBlockAdv;                           #
  
  use strict;
  use warnings FATAL => 'all';
  
  use Apache2::Filter ();
  use Apache2::RequestRec ();
  use APR::Table ();
  use Apache2::Const -compile => qw(OK);

  sub filter {                                                      #
      my $f = shift;
  
      my $data = $f->ctx;
      $data = "" unless defined $data;
  
      while ($f->read(my $buffer, 1024)) {
          $data .= $buffer if defined $buffer;
      }
  
      if ($f->seen_eos) {
          $data =~ s|<img[^>]+>||ig;
          $f->r->headers_out->unset('Content-Length');
          $f->print($data) if defined $data;
      } else {
          $f->ctx($data) if defined $data;
      }
  
      return Apache2::Const::OK;
  }
  1;

since we have buffered the data up, instead of deleting the
C-L header, we can set it to the correct value:

      if ($f->seen_eos) {
          $data =~ s|<img[^>]+>||ig;
          my $len = defined $data ? length $data : 0;
          $f->r->headers_out->set('Content-Length', $len);
          $f->print($data) if defined $data;
      } ...

Stas BekmanInput and Output Filters Slide 122



The same filter using bucket brigades

  package MyApache2::FilterTryAdBlockBB;                         #
  
  use strict;
  use warnings FATAL => 'all';
  
  use Apache2::Filter ();
  use Apache2::RequestRec ();
  use APR::Brigade ();
  use APR::Bucket ();
  use APR::Table ();
  
  use Apache2::Const -compile => qw(OK);

  sub bb_data_n_eos {                                            #
      my $bb = shift;
  
      my $seen_eos = 0;
  
      my @data;
      for (my $b = $bb->first; $b; $b = $bb->next($b)) {
          $seen_eos++, last if $b->is_eos;
          $b->read(my $bdata);
          push @data, $bdata;
      }
      return (join('', @data), $seen_eos);
  }

  sub filter {                                                    #
      my($f, $bb) = @_;
  
      my $data = $f->ctx;
      $data = "" unless defined $data;
  
      my($buffer, $seen_eos) = bb_data_n_eos($bb);
      $data .= $buffer if defined $buffer;
  
      if ($seen_eos) {
          $data =~ s|<img[^>]+>||ig;
          $f->r->headers_out->unset('Content-Length');
          $f->print($data) if defined $data;
      } else {
          $f->ctx($data) if defined $data;
      }
  
      return Apache2::Const::OK;
  }
  1;

6.4��I/O Filtering Concepts
Before getting to the real filter examples

a few concepts to help you wrap your head around ...

6.4.1��2 Ways to Manipulate Data
Apache 2.0 considers all incoming and outgoing data as
chunks of information,

disregarding their kind and source or storage methods.

These data chunks are stored in buckets, which form bucket 
brigades.

Both input and output filters, filter the data in bucket brigades.

mod_perl 2.0 filters can :

directly manipulate the bucket brigades

use a simplified streaming interface

the filter object acts similar to a filehandle,

which can be read from and printed to.

6.4.2��HTTP Request vs
Connection Filters
HTTP Request filters

are applied on:

HTTP requests body (POST) (if consumed by the content 
handler)

HTTP response body

HTTP headers are not passed through the HTTP Request 
filters.

Connections filters :

are applied at the connection level.

can modify all the incoming and outgoing data.

including HTTP headers if HTTP protocol is used

6.4.3��Multiple Invocations of
Filter Handlers

a filter gets invoked as many times as the number of bucket
brigades sent from an upstream filter or a content provider.

Response handler example, buffered STDOUT ($|==0):

  $r->print("foo");
  $r->rflush;
  $r->print("bar");

Apache generates 3 bucket brigades => 3 filter invocations

print and rflush => two buckets:

  bucket type       data
  ----------------------
  1st    transient   foo
  2nd    flush

print => one bucket:

  bucket type       data
  ----------------------
  1st    transient   bar

when the response handler returns => one bucket:

  bucket type       data
  ----------------------
  1st    eos

EOS may come as a last bucket in bb with data

Three Filter Stages :

Filter skeleton demonstrating all 3 stages :

  sub handler {
      my $f = shift;
  
      unless ($f->ctx) {  # runs on the first invocation
          init($f);
          $f->ctx(1);
      }
  
      process($f);        # runs on all invocations
  
      if ($f->seen_eos) { # runs on the last invocation
          finalize($f);
      }
  
      return Apache2::Const::OK;
  }

Stas BekmanInput and Output Filters Slide 138



Init example :

response filters changing data length have to unset the C-L
header (runs once)

  unless ($f->ctx) {
      $f->r->headers_out->unset('Content-Length');
      $f->ctx(1);
  }

Processing example :

lower the data case (runs on all invocations)

  while ($f->read(my $data, 1024)) {
      $f->print(lc $data);
  }

in complex cases may need to store data in context between
filter invocations

Finalization example :

After all data was sent out, print how many times the filter was 
invoked

  my $ctx = $f->ctx;
  $ctx->{invoked}++;
  $f->ctx($ctx); # store
  
  while ($f->read(my $data, READ_SIZE)) {
      $f->print(lc $data);
  }
  
  if ($f->seen_eos) {
     $f->print("Filter invocated $ctx->{invoked} times");
  }

Stas BekmanInput and Output Filters Slide 141



6.4.4��Blocking Calls
Data propogates in stages through the filter chain (there is no 
pipeline)

The input filter's get_brigade()  call blocks

The output filter's pass_brigade()  call blocks

Transparent input filter :

  sub in {
      my ($f, $bb, $mode, $block, $readbytes) = @_;
  
      my $rv = $f->next->get_brigade($bb, $mode, $block, $readbytes);
      return $rv unless $rv == APR::Const::SUCCESS;
  
      return Apache2::Const::OK;
  }

Transparent output filter :

  sub out {
      my ($f, $bb) = @_;
  
      my $rv = $f->next->pass_brigade($bb);                         #
      return $rv unless $rv == APR::Const::SUCCESS;
  
      return Apache2::Const::OK;
  }
  1;

6.5��mod_perl Filters
Declaration and Configuration

Now let's see how mod_perl filters are declared and 
configured.

6.5.1��Filter Priority Types
Filters are inserted into the filter chain based on their priority

mod_perl provides 2 filters types:

  Handler                  Priority           Value
  -------------------------------------------------
  FilterRequestHandler     AP_FTYPE_RESOURCE    10
  FilterConnectionHandler  AP_FTYPE_PROTOCOL    30

Apache provides others as well

6.5.2��PerlInputFilterHandler
The PerlInputFilterHandler  handler:

registers a filter for input filtering

and inserts it into the input filters chain.

automatically AutoLoaded:

  PerlInputFilterHandler MyApache2::FilterTest
  PerlInputFilterHandler MyApache2::FilterTest::lc

avoid autoloading:

  PerlInputFilterHandler -MyApache2::FilterTest::lc

6.5.3��PerlOutputFilterHandler
The PerlOutputFilterHandler  handler:

registers an output filter

and inserts it into the output filters chain.

automatically AutoLoaded like PerlInputFilterHandler  

6.5.4��PerlSetInputFilter
mod_perl equivalent for SetInputFilter

makes sure that mod_perl and non-mod_perl input filters of
the same priority get inserted in the order they are configured

For example:

  PerlSetInputFilter FILTER_FOO
  PerlInputFilterHandler MyApache2::FilterInputFoo

       response handler
             /\
         FILTER_FOO
             /\
   MyApache2::FilterInputFoo
             /\
     core input filters
             /\
           network

6.5.5��PerlSetOutputFilter
mod_perl equivalent for SetOutputFilter

makes sure that mod_perl and non-mod_perl output filters of
the same priority get inserted in the order they are configured

For example:

  PerlSetOutputFilter INCLUDES
  PerlOutputFilterHandler MyApache2::FilterOutputFoo

      response handler
             \/
          INCLUDES
             \/
   MyApache2::FilterOutputFoo
             \/
     core output filters
             \/
           network

If filters have different priorities the insertion order might be 
different:

  PerlSetOutputFilter DEFLATE
  PerlSetOutputFilter INCLUDES
  PerlOutputFilterHandler MyApache2::FilterOutputFoo

      response handler
             \/
          INCLUDES
             \/
   MyApache2::FilterOutputFoo
             \/
           DEFLATE
             \/
     core output filters
             \/
           network

6.5.6��Adding OutputFilters 
Dynamically

one can push filter callbacks at request time

e.g. add an output filter at fixup phase

  <Files *\.html >
    PerlFixupHandler MyApache2::AddFilterDyn
  </Files>

  package MyApache2::AddFilterDyn;
  
  use Apache2::Filter;
  use MyApache2::FilterObfuscate;
  
  use Apache2::Const -compile => qw(OK);
  
  sub handler {

      my $r = shift;
  
      $r->add_output_filter(\&MyApache2::FilterObfuscate::handler);
  
      return Apache2::Const::OK;
  }
  
  1;

6.5.7��HTTP Request vs.
Connection Filters

method attributes set the filter type:

  sub handler : FilterRequestHandler    { ... }
  sub handler : FilterConnectionHandler { ... }

HTTP Request filter handlers are declared using the 
FilterRequestHandler  attribute.

HTTP request input and output filters skeleton:

  package MyApache2::FilterRequestFoo;
  use base qw(Apache2::Filter);
  
  sub input  : FilterRequestHandler {
      my($f, $bb, $mode, $block, $readbytes) = @_;
      #...
  }
  
  sub output : FilterRequestHandler {
      my($f, $bb) = @_;
      #...
  }
  1;

If the attribute is not specified, the default 
FilterRequestHandler  attribute is assumed.

Filters specifying subroutine attributes must sub-class 
Apache2::Filter , others only need to:

  use Apache2::Filter ();

Request filters are usually configured in the <Location>  or
equivalent sections:

  PerlModule MyApache2::FilterRequestFoo
  PerlModule MyApache2::NiceResponse
  <Location /filter_foo>
      SetHandler modperl
      PerlResponseHandler     MyApache2::NiceResponse
      PerlInputFilterHandler  MyApache2::FilterRequestFoo::input
      PerlOutputFilterHandler MyApache2::FilterRequestFoo::output
  </Location>

Connection filter handlers are declared using the 
FilterConnectionHandler  attribute.

Connection input and output filters skeleton:

  package MyApache2::FilterConnectionBar;
  use base qw(Apache2::Filter);
  
  sub input  : FilterConnectionHandler {
      my($f, $bb, $mode, $block, $readbytes) = @_;
      #...
  }
  
  sub output : FilterConnectionHandler {
      my($f, $bb) = @_;
      #...
  }
  1;

This time the configuration must be done outside the 
<Location>  or equivalent sections, 

usually within the <VirtualHost>  or the global server 
configuration:

  Listen 8005
  <VirtualHost _default_:8005>
      PerlModule MyApache2::FilterConnectionBar
      PerlModule MyApache2::NiceResponse
   
      PerlInputFilterHandler  MyApache2::FilterConnectionBar::input
      PerlOutputFilterHandler MyApache2::FilterConnectionBar::output
      <Location />
          SetHandler modperl
          PerlResponseHandler MyApache2::NiceResponse
      </Location>
   
  </VirtualHost>

For HTTP requests the only difference between connection
filters and request filters is that the former see everything: the
headers and the body, whereas the latter see only the body.

mod_perl provides two interfaces to filtering:

a direct mapping to buckets and bucket brigades

and a simpler, stream-oriented interface

Following examples will explain the difference

6.5.8��Filter Initialization Phase
There is a special attribute duet 
FilterHasInitHandler /FilterInitHandler  but it's left
for your offline reading.

6.6��All-in-One Filter
The MyApache2::FilterSnoop  handler silently snoops on
the data that goes through request and connection filters, in
the input and output modes.

First let's develop a simple response handler that simply
dumps the request's args and content as strings:

  package MyApache2::Dump;
  
  use strict;
  use warnings FATAL => 'all';
  
  use Apache2::RequestRec ();
  use Apache2::RequestIO ();
  use Apache2::Filter ();
  use APR::Brigade ();

  use APR::Bucket ();
  
  use Apache2::Const -compile => qw(OK M_POST);
  
  sub handler {
      my $r = shift;
      $r->content_type('text/plain');
  
      $r->print("args:\n", $r->args, "\n");
  
      if ($r->method_number == Apache2::Const::M_POST) {
          my $data = content($r);
          $r->print("content:\n$data\n");
      }
  
      return Apache2::Const::OK;
  }

  sub content {
     
     # see your handouts, the details are unimportant here
     
  }
  
 1;

which is configured as:

  PerlModule MyApache2::Dump
  <Location /dump>
      SetHandler modperl
      PerlResponseHandler MyApache2::Dump
  </Location>

If we issue the following request:

  % echo "mod_perl rules" | POST 'http://localhost:8002/dump?foo=1&bar=2'

the response will be:

  args:
  foo=1&bar=2
  content:
  mod_perl rules

As you can see it simply dumped the query string and the
posted data.

Now let's write the snooping filter:

  package MyApache2::FilterSnoop;
  
  use strict;
  use warnings;
  
  use base qw(Apache2::Filter);
  use Apache2::FilterRec ();
  use APR::Brigade ();
  use APR::Bucket ();
  use APR::BucketType ();
  
  use Apache2::Const -compile => qw(OK DECLINED);
  use APR::Const     -compile => ':common';
  
  sub connection : FilterConnectionHandler { snoop("connection", @_) }
  sub request    : FilterRequestHandler    { snoop("request",    @_) }

  sub snoop {
      my $type = shift;
      my($f, $bb, $mode, $block, $readbytes) = @_; # filter args
  
      # $mode, $block, $readbytes are passed only for input filters
      my $stream = defined $mode ? "input" : "output";
  
      # read the data and pass-through the bucket brigades unchanged
      if (defined $mode) {
          # input filter
          my $rv = $f->next->get_brigade($bb, $mode, $block, $readbytes);
          return $rv unless $rv == APR::Const::SUCCESS;
          bb_dump($type, $stream, $bb);
      }
      else {
          # output filter
          bb_dump($type, $stream, $bb);
          my $rv = $f->next->pass_brigade($bb);
          return $rv unless $rv == APR::Const::SUCCESS;
      }
  
      return Apache2::Const::OK;
  }

  sub bb_dump {
      my($type, $stream, $bb) = @_;
  
      my @data;
      for (my $b = $bb->first; $b; $b = $bb->next($b)) {
          $b->read(my $bdata);
          push @data, $b->type->name, $bdata;
      }
  
      # send the sniffed info to STDERR so not to interfere with normal
      # output
      my $direction = $stream eq 'output' ? ">>>" : "<<<";
      print STDERR "\n$direction $type $stream filter\n";
  
      my $c = 1;
      while (my($btype, $data) = splice @data, 0, 2) {
          print STDERR "    o bucket $c: $btype\n";
          print STDERR "[$data]\n";
          $c++;
      }
  }
  1;

We want to use this somewhat complicated filter to visualize
how various kind of filters work.

Since this module combines several kind of filters it's the best
to try to understand its implementation after you understand
standalone filters' implementation.

At this stage what's important is to see it at work.

Let's snoop on connection and request filter levels in both
directions by applying the following configuration:

  Listen 8008
  <VirtualHost _default_:8008>
      PerlModule MyApache2::FilterSnoop
      PerlModule MyApache2::Dump
  
      # Connection filters
      PerlInputFilterHandler  MyApache2::FilterSnoop::connection
      PerlOutputFilterHandler MyApache2::FilterSnoop::connection
  
      <Location /dump>
          SetHandler modperl
          PerlResponseHandler MyApache2::Dump
          # Request filters
          PerlInputFilterHandler  MyApache2::FilterSnoop::request
          PerlOutputFilterHandler MyApache2::FilterSnoop::request
      </Location>
  
  </VirtualHost>

If we issue the following request:

  % echo "mod_perl rules" | POST 'http://localhost:8008/dump?foo=1&bar=2'

the response doesn't get affected by the snooping filter

though we can see all the diagnostics in error_log

First we can see the connection input filter at work, as it
processes the HTTP headers.

We can see that for this request each header is put into a
separate brigade with a single bucket.

The data is conveniently enclosed by []  so you can see the
new line characters as well.

  <<< connection input filter
      o bucket 1: HEAP
  [POST /dump?foo=1&bar=2 HTTP/1.1
  ]
  
  <<< connection input filter
      o bucket 1: HEAP
  [TE: deflate,gzip;q=0.3
  ]

Stas BekmanInput and Output Filters Slide 175



  
  <<< connection input filter
      o bucket 1: HEAP
  [Connection: TE, close
  ]
  
  <<< connection input filter
      o bucket 1: HEAP
  [Host: localhost:8008
  ]
  
  <<< connection input filter
      o bucket 1: HEAP
  [User-Agent: lwp-request/2.01
  ]
  
  <<< connection input filter
      o bucket 1: HEAP

  [Content-Length: 14
  ]
  
  <<< connection input filter
      o bucket 1: HEAP
  [Content-Type: application/x-www-form-urlencoded
  ]
  
  <<< connection input filter
      o bucket 1: HEAP
  [
  ]

Here the HTTP header has been terminated by a double new 
line.

So far all the buckets were of the HEAP type, meaning that
they were allocated from the heap memory.

Notice that the HTTP request input filters will never see the
bucket brigades with HTTP headers, as it has been consumed by
the last core connection filter.

The following two entries are generated when 
MyApache2::Dump::handler  reads the POSTed content:

  <<< connection input filter
      o bucket 1: HEAP
  [mod_perl rules]
  
  <<< request input filter
      o bucket 1: HEAP
  [mod_perl rules]
      o bucket 2: EOS
  []

The connection input filter is run before the request input filter.

Both filters see the same data, since they don't modify it

The bucket of type EOS indicates the end of stream.

Next we can see that MyApache2::Dump::handler  has
generated its response.

However only the request output filter is filtering it at this 
point:

  >>> request output filter
      o bucket 1: TRANSIENT
  [args:
  foo=1&bar=2
  content:
  mod_perl rules
  ]

Next, Apache injects the HTTP headers:

  >>> connection output filter
      o bucket 1: HEAP
  [HTTP/1.1 200 OK
  Date: Fri, 04 Jun 2004 09:13:26 GMT
  Server: Apache/2.0.50-dev (Unix) mod_perl/1.99_15-dev
  Perl/v5.8.4 mod_ssl/2.0.50-dev OpenSSL/0.9.7c DAV/2
  Connection: close
  Transfer-Encoding: chunked
  Content-Type: text/plain; charset=ISO-8859-1
  
  ]

the first response body's brigade inside the connection output 
filter:

  >>> connection output filter
      o bucket 1: TRANSIENT
  [2b
  ]
      o bucket 2: TRANSIENT
  [args:
  foo=1&bar=2
  content:
  mod_perl rules
  
  ]
      o bucket 3: IMMORTAL
  [
  ]

Finally the output is flushed, 

to make sure that any buffered output is sent to the client:

  >>> connection output filter
      o bucket 1: IMMORTAL
  [0
  
  ]
      o bucket 2: EOS
  []

This module helps to understand that each filter handler can
be called many time during each request and connection.

It's called for each bucket brigade.

the HTTP request input filter is called only if there is some
POSTed data to read

If you run the same request without POSTing any data or
simply running a GET request, the request input filter won't be 
called.

6.7��Input Filters
mod_perl supports 2 kinds of input filters:

1.  Connection input filters

2.  HTTP Request input filters

6.7.1��Connection Input Filters
Let's write a poor man's s/GET/HEAD/ rewrite handler

The handler looks for the data like:

  GET /perl/test.pl HTTP/1.1

and turns it into:

  HEAD /perl/test.pl HTTP/1.1

  package MyApache2::InputFilterGET2HEAD;                            #
  
  use strict;
  use warnings;
  
  use base qw(Apache2::Filter);
  
  use APR::Brigade ();
  use APR::Bucket ();
  
  use Apache2::Const -compile => 'OK';
  use APR::Const     -compile => ':common';

  sub handler : FilterConnectionHandler {                                   #
      my($f, $bb, $mode, $block, $readbytes) = @_;
  
      return Apache2::Const::DECLINED if $f->ctx;
  
      my $rv = $f->next->get_brigade($bb, $mode, $block, $readbytes);
      return $rv unless $rv == APR::Const::SUCCESS;
  
      for (my $b = $bb->first; $b; $b = $bb->next($b)) {
          $b->read(my $data);
          warn("data: $data\n");
  
          if ($data and $data =~ s|^GET|HEAD|) {
              my $nb = APR::Bucket->new($bb->bucket_alloc, $data);
              $b->insert_after($nb);
              $b->remove; # no longer needed
              $f->ctx(1); # flag that that we have done the job
              last;
          }
      }
  
      Apache2::Const::OK;
  }
  1;

For example, consider the following response handler:

  package MyApache2::RequestType;                              #
  
  use Apache2::RequestIO ();
  use Apache2::RequestRec ();
  use Apache2::Response ();
  
  use Apache2::Const -compile => 'OK';
  
  sub handler {
      my $r = shift;
  
      $r->content_type('text/plain');
      my $response = "the request type was " . $r->method;
      $r->set_content_length(length $response);
      $r->print($response);
  
      Apache2::Const::OK;
  }
  
  1;

which returns to the client the request type it has issued.

In the case of the HEAD request Apache will discard the
response body, but it'll will still set the correct 
Content-Length  header, which will be 24 in case of the 
GET request and 25 for HEAD.

Therefore if this response handler is configured as:

  Listen 8005
  <VirtualHost _default_:8005>
      <Location />
          SetHandler modperl
          PerlResponseHandler +MyApache2::RequestType
      </Location>
  </VirtualHost>

and a GET request is issued to /:

  panic% perl -MLWP::UserAgent -le \
  '$r = LWP::UserAgent->new()->get("http://localhost:8005/"); \
  print $r->headers->content_length . ": ".  $r->content'
  24: the request type was GET

where the response's body is:

  the request type was GET

And the Content-Length  header is set to 24.

However if we enable the 
MyApache2::InputFilterGET2HEAD  input connection 
filter:

  Listen 8005
  <VirtualHost _default_:8005>
      PerlInputFilterHandler +MyApache2::InputFilterGET2HEAD
  
      <Location />
          SetHandler modperl
          PerlResponseHandler +MyApache2::RequestType
      </Location>
  </VirtualHost>

And issue the same GET request, we get only:

  25:

Stas BekmanInput and Output Filters Slide 194



which means that the body was discarded by Apache,
because our filter turned the GET request into a HEAD request.

If Apache wasn't discarding the body on HEAD, the response
would be:

  the request type was HEAD

that's why the content length is reported as 25 and not 24 as
in the real GET request.

6.7.2��HTTP Request Input Filters
Request filters are similar to connection filters, but have an
access to a request object and they don't see the headers.

6.7.3��Bucket Brigade-based Input 
Filters

Here is the request input filter that lowercases the request's 
body:

  package MyApache2::InputRequestFilterLC;                     #
  
  use strict;
  use warnings;
  
  use base qw(Apache2::Filter);
  
  use Apache2::Connection ();
  use APR::Brigade ();
  use APR::Bucket ();
  
  use Apache2::Const -compile => 'OK';
  use APR::Const     -compile => ':common';

  sub handler : FilterRequestHandler {
      my($f, $bb, $mode, $block, $readbytes) = @_;
  
      my $c = $f->c;
      my $bb_ctx = APR::Brigade->new($c->pool, $c->bucket_alloc);
      my $rv = $f->next->get_brigade($bb_ctx, $mode, $block, $readbytes);
      return $rv unless $rv == APR::Const::SUCCESS;
  
      while (!$bb_ctx->is_empty) {
          my $b = $bb_ctx->first;
  
          if ($b->is_eos) {
              $bb->insert_tail($b);
              last;
          }
  
          my $len = $b->read(my $data);
          $b = APR::Bucket->new($bb->bucket_alloc, lc $data) if $len;
  
          $b->remove;
          $bb->insert_tail($b);
      }
  

Stas BekmanInput and Output Filters Slide 198



      Apache2::Const::OK;
  }
  
  1;

the MyApache2::Dump  response handler dumps the query
string and the content body as a response

Using the following configuration:

  <Location /lc_input>
      SetHandler modperl
      PerlResponseHandler    +MyApache2::Dump
      PerlInputFilterHandler +MyApache2::InputRequestFilterLC
  </Location>

When issuing a POST request:

 % echo "mOd_pErl RuLeS" | POST 'http://localhost:8002/lc_input?FoO=1&BAR=2'

we get a response:

  args:
  FoO=1&BAR=2
  content:
  mod_perl rules

Indeed we can see that our filter has lowercased the POSTed
body, before the content handler received it.

The query string didn't change, since the filter didn't operate
on headers.

6.7.4��Stream-oriented Input 
Filters
  package MyApache2::InputRequestFilterLC2;                      #
  
  use base qw(Apache2::Filter);
  
  use Apache2::Const -compile => 'OK';
  
  sub handler : FilterRequestHandler {
      my $f = shift;
  
      while ($f->read(my $buffer, 1024)) {
          $f->print(lc $buffer);
      }
  
      Apache2::Const::OK;
  }
  1;

Configuration:

 <Location /lc_input2>
      SetHandler modperl
      PerlResponseHandler    +MyApache2::Dump
      PerlInputFilterHandler +MyApache2::InputRequestFilterLC2
  </Location>

We get the same results as from using the bucket brigades 
filter

6.8��Output Filters
mod_perl supports 2 kinds of output filters:

1.  Connection input filters

2.  HTTP Request input filters

6.8.1��Connection Output Filters
Connection filters are similar to HTTP Request filters,

but can see all  the data that is going through the server.

We will concentrate on HTTP request filters

6.8.2��HTTP Request Output 
Filters

Output filters can be written using the bucket brigades
manipulation or the simplified stream-oriented interface.

Here is a response handler that send two lines of output in a
single string:

1.  numerals: 1234567890

2.  the English alphabet

  package MyApache2::SendAlphaNum;                    #
  
  use Apache2::RequestRec ();
  use Apache2::RequestIO ();
  
  use Apache2::Const -compile => qw(OK);

  sub handler {
      my $r = shift;
  
      $r->content_type('text/plain');
  
      $r->print(1..9, "0\n");
      $r->print('a'..'z', "\n");
  
      Apache2::Const::OK;
  }
  1;

Stas BekmanInput and Output Filters Slide 207



The purpose of our request output filter is to reverse every
line of the response, preserving the new line characters in
their places.

Since we want to reverse characters only in the response
body we will use the HTTP request output filter.

6.8.3��Stream-oriented Output 
Filter
  package MyApache2::FilterReverse1;                          #
  
  use Apache2::Filter ();
  use Apache2::Const -compile => qw(OK);
  
  sub handler : FilterRequestHandler {
      my $f = shift;
  
      while ($f->read(my $buffer, 1024)) {
          for (split "\n", $buffer) {
              $f->print(scalar reverse $_);
              $f->print("\n");
          }
      }
      Apache2::Const::OK;
  }
  1;

Add the following configuration to httpd.conf:

  PerlModule MyApache2::FilterReverse1
  PerlModule MyApache2::SendAlphaNum
  <Location /reverse1>
      SetHandler modperl
      PerlResponseHandler     MyApache2::SendAlphaNum
      PerlOutputFilterHandler MyApache2::FilterReverse1
  </Location>

Now when a request to /reverse1 is made, the response
handler MyApache2::SendAlphaNum::handler()  sends:

  1234567890
  abcdefghijklmnopqrstuvwxyz

as a response and the output filter handler 
MyApache2::FilterReverse1::handler  reverses the
lines, so the client gets:

  0987654321
  zyxwvutsrqponmlkjihgfedcba

Stas BekmanInput and Output Filters Slide 211



6.8.4��Bucket Brigade-based
Output Filters

The bucket brigades filtering API implementation:

  package MyApache2::FilterReverse2;                                   #
  
  use strict;
  use warnings;
  
  use base qw(Apache2::Filter);
  
  use APR::Brigade ();
  use APR::Bucket ();
  
  use Apache2::Const -compile => 'OK';
  use APR::Const -compile => ':common';
  
  sub handler : FilterRequestHandler {
      my($f, $bb) = @_;

      my $bb_ctx = APR::Brigade->new($f->c->pool, $f->c->bucket_alloc);   #
  
      while (!$bb->is_empty) {
          my $b = $bb->first;
          $b->remove;
  
          if ($b->is_eos) {
              $bb_ctx->insert_tail($b);
              last;
          }
  
          if ($b->read(my $data)) {
              $data = join "",
                  map {scalar(reverse $_), "\n"} split "\n", $data;
              $b = APR::Bucket->new($bb->bucket_alloc, $data);
          }
          $bb_ctx->insert_tail($b);
      }
  
      my $rv = $f->next->pass_brigade($bb_ctx);
      return $rv unless $rv == APR::Const::SUCCESS;
  
      Apache2::Const::OK;
  }
  1;

And the corresponding configuration:

  PerlModule MyApache2::FilterReverse2
  PerlModule MyApache2::SendAlphaNum
  <Location /reverse2>
      SetHandler modperl
      PerlResponseHandler     MyApache2::SendAlphaNum
      PerlOutputFilterHandler MyApache2::FilterReverse2
  </Location>

Now when a request to /reverse2 is made, the client gets:

  0987654321
  zyxwvutsrqponmlkjihgfedcba

as expected.

7�� HTTP Handlers 

7.1��About
HTTP Request Cycle Phases

PerlPostReadRequestHandler

PerlTransHandler

PerlMapToStorageHandler

PerlHeaderParserHandler

PerlInitHandler

PerlAccessHandler

PerlAuthenHandler

PerlAuthzHandler

PerlTypeHandler

PerlFixupHandler

PerlResponseHandler

PerlLogHandler

PerlCleanupHandler

7.2��HTTP Request Handler 
Skeleton

All HTTP Request handlers have the following structure:

  package MyApache2::MyHandlerName;
  
  # load modules that are going to be used
  use ...;
  
  # compile (or import) constants
  use Apache2::Const -compile => qw(OK);
  
  sub handler {
      my $r = shift;
      
      # handler code comes here

      return Apache2::Const::OK; # or another status constant
  }
  1;

If the handler is declared as a method handler:

  sub handler : method {
      my($class, $r) = @_;

7.3��HTTP Request Cycle 
Phases

Almost identical to mod_perl 1.0. The differences are:

new phase: PerlMapToStorageHandler

rename: PerlHandler  => PerlResponseHandler

PerlResponseHandler  now includes filtering

Here is the HTTP request life cycle in mod_perl 2.0:

an HTTP request is processed by 12 phases:

1.  PerlPostReadRequestHandler (PerlInitHandler)

2.  PerlTransHandler

3.  PerlMapToStorageHandler

4.  PerlHeaderParserHandler (PerlInitHandler)

5.  PerlAccessHandler

6.  PerlAuthenHandler

7.  PerlAuthzHandler

8.  PerlTypeHandler

9.  PerlFixupHandler

10.  PerlResponseHandler

11.  PerlLogHandler

12.  PerlCleanupHandler

7.3.1��PerlPostReadRequestHandler

Stas BekmanHTTP Handlers Slide 227



Stas BekmanHTTP Handlers Slide 228



The post_read_request phase is the first request phase and
happens immediately after the request has been read and
HTTP headers were parsed.

This phase is usually used to do processing that must happen
once per request.

For example Apache2::Reload  is usually invoked at this
phase to reload modified Perl modules.

The following ModPerl::Registry  script prints when the
last time httpd.conf has been modified, compared to the start
of the request process time:

  use strict;
  use warnings;
  
  use Apache2::ServerUtil ();
  use Apache2::RequestIO ();
  use File::Spec::Functions qw(catfile);
  
  my $r = shift;
  $r->content_type('text/plain');
  
  my $conf_file = catfile Apache2::ServerUtil::server_root,
      "conf", "httpd.conf";
  
  printf "$conf_file is %0.2f minutes old\n", 60*24*(-M $conf_file);

The script reports incorrect time most of the time

Because -M reports the difference between file's modification
time and $^T

Under mod_perl $^T  is set when the process starts and
doesn't change after that.

Solution: Reset $^T  on each request

  package MyApache2::TimeReset;
  
  use strict;
  use warnings;
  
  use Apache2::RequestRec ();
  
  use Apache2::Const -compile => 'OK';
  
  sub handler {
      my $r = shift;
      $^T = $r->request_time;
      return Apache2::Const::OK;
  }
  1;

Stas BekmanHTTP Handlers Slide 232



We could do:

  $^T = time();

But $r->request_time  already stores the request's start
time, so we get it without performing an additional system call.

To enable it just add to httpd.conf:

  PerlPostReadRequestHandler MyApache2::TimeReset

either to the global section, or to the <VirtualHost>  section if
you want this handler to be run only for a specific virtual host.

7.3.2��PerlTransHandler

The translate phase is used to manipulate a request's URI

If no custom handler is provided, the server's standard
translation rules (e.g., Alias  directives, mod_rewrite, etc.)
will continue to be used.

Also used to modify the URI itself and the request method.

This is also a good place to register new handlers for the
following phases based on the URI.

mod_rewrite ala mod_perl :

e.g. if the previously static pages are now autogenerated, and

  http://example.com/news/20021031/09/index.html                    _

is now handled by:

  http://example.com/perl/news.pl?date=20021031;id=09;page=index.html

we can keep using the old URI, transparent to news.pl and 
users

  package MyApache2::RewriteURI;
  
  use strict;
  use warnings;
  
  use Apache2::RequestRec ();
  
  use Apache2::Const -compile => qw(DECLINED);
  
  sub handler {
      my $r = shift;
  
      my($date, $id, $page) = $r->uri =~ m|^/news/(\d+)/(\d+)/(.*)|;
      $r->uri("/perl/news.pl");
      $r->args("date=$date;id=$id;page=$page");
  
      return Apache2::Const::DECLINED;
  }
  1;

To configure this module simply add to httpd.conf:

  PerlTransHandler +MyApache2::RewriteURI

7.3.3��PerlMapToStorageHandler

The map_to_storage phase is used to translate request's URI
into a corresponding filename

If no custom handler is provided, the server will try to walk the
filesystem trying to find what file or directory corresponds to
the request's URI.

most likely you want to shortcut this phase for mod_perl

The Don't-waste-time handler:

  package MyApache2::NoTranslation;                                  #
  
  use strict;
  use warnings FATAL => 'all';
  
  use Apache2::Const -compile => qw(OK);
  
  sub handler {
      my $r = shift;
  
      # skip ap_directory_walk stat() calls
      return Apache2::Const::OK;
  }
  1;

config:

  PerlMapToStorageHandler MyApache2::NoTranslation

But this can be done from httpd.conf too!

  PerlMapToStorageHandler Apache2::Const::OK

If you haven't already compiled Apache2::Const::OK
elsewhere, you should add:

  <Perl>
      use Apache2::Const -compile => qw(OK);
  </Perl>

same handler but support TRACE calls

  package MyApache2::NoTranslation2;                                 #
  
  use strict;
  use warnings FATAL => 'all';
  
  use Apache2::RequestRec ();

  use Apache2::Const -compile => qw(DECLINED OK M_TRACE);
  
  sub handler {
      my $r = shift;
  
      return Apache2::Const::DECLINED
          if $r->method_number == Apache2::Const::M_TRACE;
  
      # skip ap_directory_walk stat() calls
      return Apache2::Const::OK;
  }
  1;

Another way to prevent the core translation:

  $r->filename(__FILE__);

this can be done in PerlTransHandler  as well

7.3.4��PerlHeaderParserHandler

The header_parser phase is the first phase to happen after
the request has been mapped to its <Location>  (or an
equivalent container).

At this phase the handler can examine the request headers
and to take a special action based on these.

e.g., this phase can be used to block evil clients targeting
certain resources, while few resources were wasted so far.

Apache handles the HEAD, GET, POST and several other
HTTP methods.

But you can invent your own methods and make Apache
accept them.

e.g., emails are very similar to HTTP messages: they have a
set of headers and a body, sometimes a multi-part body.

MyApache2::SendEmail  extends HTTP by adding a
support for the EMAIL method.

It enables the new extension and pushes the real content
handler during the PerlHeaderParserHandler  phase:

  <Location /email>
      PerlHeaderParserHandler MyApache2::SendEmail
  </Location>

  package MyApache2::SendEmail;                                      #
  
  use strict;
  use warnings;
  
  use Apache2::RequestRec ();
  use Apache2::RequestIO ();
  use Apache2::RequestUtil ();
  use Apache2::Server ();
  use Apache2::ServerRec ();
  use Apache2::Process ();
  use APR::Table ();
  
  use Apache2::Const -compile => qw(DECLINED OK);
  
  use constant SMTP_HOSTNAME => "localhost";

  sub handler {                                                      #
      my $r = shift;
      return Apache2::Const::DECLINED unless $r->method eq 'EMAIL';
  
      $r->server->method_register('EMAIL');
      $r->handler("perl-script");
      $r->push_handlers(PerlResponseHandler => \&send_email_handler);
  
      return Apache2::Const::OK;
  }
  
  sub send_email_handler {                                           #
      my $r = shift;
  
      my %headers = map {$_ => $r->headers_in->get($_)} 
          qw(To From Subject);
      my $content = content($r);
  
      my $status = send_email(\%headers, \$content);
  
      $r->content_type('text/plain');
      $r->print($status ? "ACK" : "NACK");
      return Apache2::Const::OK;
  }

  sub send_email {                                                  #
      my($rh_headers, $r_body) = @_;
  
      require MIME::Lite;
      MIME::Lite->send("smtp", SMTP_HOSTNAME, Timeout => 60);
      my $msg = MIME::Lite->new(%$rh_headers, Data => $$r_body);
  
      $msg->send;
  }
  
  sub content {                                                     #
      my $r = shift;
  
      # unimportant here, see your handouts
  
      return $buf;
  }
  1;

when you extend an HTTP protocol you need to have a client
that knows how to use the extension.

So here is a simple client that uses LWP::UserAgent  to
issue an EMAIL method request over HTTP protocol:

  require LWP::UserAgent;                                           #
  
  my $url = "http://localhost:8000/email/";
  
  my %headers = (
      From    => 'example@example.com',
      To      => 'example@example.com',
      Subject => '3 weeks in Tibet',
  );
  
  my $content = <<EOI;
  I didn't have an email software,
  but could use HTTP so I'm sending it over HTTP
  EOI
  
  my $headers = HTTP::Headers->new(%headers);
  my $req = HTTP::Request->new("EMAIL", $url, $headers, $content);
  my $res = LWP::UserAgent->new->request($req);
  print $res->is_success ? $res->content : "failed";

7.3.5��PerlInitHandler

If configured inside any container directive, except 
<VirtualHost> , this handler is an alias for 
PerlHeaderParserHandler .

It is the first handler to be invoked when serving a request.

For example if Apache2::Reload  is configured as:

  PerlInitHandler Apache2::Reload
  PerlSetVar ReloadAll Off
  PerlSetVar ReloadModules "MyApache2::*"

then it will always monitor and reload all MyApache2::*
modules that have been modified since the last HTTP 
request.

However if we move the global configuration into a 
<Location>  container:

  <Location /devel>
      PerlInitHandler Apache2::Reload
      PerlSetVar ReloadAll Off
      PerlSetVar ReloadModules "MyApache2::*"
      SetHandler perl-script
      PerlResponseHandler ModPerl::Registry
      Options +ExecCGI
  </Location>

it acts as an alias for PerlPostReadRequestHandler

therefore Apache2::Reload  will reload the modified
modules, only when a request to the /devel namespace is 
made

7.3.6��PerlAccessHandler

The access_checker phase is the first of three handlers that
are involved in what's known as AAA: Authentication and
Authorization, and Access control.

This phase can be used to restrict access from a certain IP
address, time of the day or any other rule not connected to
the user's identity.

The concept behind access checker handler is very simple:

return Apache2::Const::FORBIDDEN  if the access is
not allowed,

otherwise return Apache2::Const::OK .

This handler denies requests made from IPs on the blacklist:

  package MyApache2::BlockByIP;                                      #
  
  use Apache2::RequestRec ();
  use Apache2::Connection ();
  
  use Apache2::Const -compile => qw(FORBIDDEN OK);
  
  my %bad_ips = map {$_ => 1} qw(127.0.0.1 10.0.0.4);
  
  sub handler {
      my $r = shift;
      return exists $bad_ips{$r->connection->remote_ip}
          ? Apache2::Const::FORBIDDEN
          : Apache2::Const::OK;
  }
  1;

To enable the handler simply add it to the container that
needs to be protected.

For example to protect an access to the registry scripts
executed from the base location /perl add:

  <Location /perl/>
      SetHandler perl-script
      PerlResponseHandler ModPerl::Registry
      PerlAccessHandler MyApache2::BlockByIP
      Options +ExecCGI
  </Location>

7.3.7��PerlAuthenHandler

The check_user_id (authen) phase is called whenever the
requested file or directory is password protected.

This, in turn, requires that the directory be associated with 
AuthName, AuthType  and at least one require  directive.

This phase is usually used to verify a user's identification 
credentials.

If the credentials are verified to be correct, the handler should
return Apache2::Const::OK .

to defer the decision: Apache2::Const::DECLINED

Otherwise the handler returns 
Apache2::Const::HTTP_UNAUTHORIZED  to indicate that
the user has not authenticated successfully.

When Apache sends the HTTP header with this code, the
browser will normally pop up a dialog box that prompts the user for
login information.

The following example handler verifies that:

  SECRET_LENGTH == length join " ", $username, $password;

  package MyApache2::SecretLengthAuth;                               #
  
  use Apache2::Access ();
  use Apache2::RequestUtil ();
  
  use Apache2::Const -compile => qw(OK DECLINED HTTP_UNAUTHORIZED);
  
  use constant SECRET_LENGTH => 14;
  
  sub handler {
      my $r = shift;
  
      my ($status, $password) = $r->get_basic_auth_pw;
      return $status unless $status == Apache2::Const::OK;
  
      return Apache2::Const::OK 
          if SECRET_LENGTH == length join " ", $r->user, $password;
  
      $r->note_basic_auth_failure;
      return Apache2::Const::HTTP_UNAUTHORIZED;
  }
  1;

To enable this handler you have to tell Apache:

what authentication scheme to use (AuthType : Basic
or Digest )

what's the authentication realm (AuthName: any string)

the Require  directive is needed to specify which
usernames are allowed to authenticate. If you set it to 
valid-user  any username will do.

  <Location /perl/>
      SetHandler perl-script
      PerlResponseHandler ModPerl::Registry
      PerlAuthenHandler MyApache2::SecretLengthAuth
      Options +ExecCGI
  
      AuthType Basic
      AuthName "The Gate"
      Require valid-user
  </Location>

7.3.8��PerlAuthzHandler

The auth_checker (authz) phase is used for authorization 
control.

This phase requires a successful authentication from the
previous phase, because a username is needed in order to
decide whether a user is authorized to access the requested 
resource.

As this phase is tightly connected to the authentication phase,
the handlers registered for this phase are only called when
the requested resource is password protected, similar to the
auth phase.

The handler is expected to return:

Apache2::Const::DECLINED  to defer the decision,

Apache2::Const::OK  to indicate its acceptance of the
user's authorization,

or Apache2::Const::HTTP_UNAUTHORIZED  to
indicate that the user is not authorized to access the requested 
document.

The MyApache2::SecretResourceAuthz  handler grants
access to certain resources only to certain users who have
already properly authenticated:

  package MyApache2::SecretResourceAuthz;                            #
  
  use strict;
  use warnings;
    
  use Apache2::Access ();
  use Apache2::RequestUtil ();
  
  use Apache2::Const -compile => qw(OK HTTP_UNAUTHORIZED);
  
  my %protected = (
      'admin'  => ['stas'],
      'report' => [qw(stas boss)],
  );

  sub handler {                                                     #
      my $r = shift;
  
      my $user = $r->user;
      if ($user) {
          my($section) = $r->uri =~ m|^/company/(\w+)/|;
          if (defined $section && exists $protected{$section}) {
              my $users = $protected{$section};
              return Apache2::Const::OK if grep { $_ eq $user } @$users;
          }
          else {
              return Apache2::Const::OK;
          }
      }
  
      $r->note_basic_auth_failure;
      return Apache2::Const::HTTP_UNAUTHORIZED;
  }
  
  1;

The configuration is similar to PerlAuthenHandler , this
time we just add the PerlAuthzHandler  setting. The rest
doesn't change.

  Alias /company/ /home/httpd/httpd-2.0/perl/
  <Location /company/>
      SetHandler perl-script
      PerlResponseHandler ModPerl::Registry
      PerlAuthenHandler   MyApache2::SecretLengthAuth
      PerlAuthzHandler    MyApache2::SecretResourceAuthz
      Options +ExecCGI
  
      AuthType Basic
      AuthName "The Secret Gate"
      Require valid-user
  </Location>

7.3.9��PerlTypeHandler

The type_checker phase is used to set the response MIME
type (Content-type ) and sometimes other bits of document
type information like the document language.

e.g., mod_autoindex , which performs automatic directory
indexing, uses this phase to map the filename extensions to
the corresponding icons which will be later used in the listing
of files.

Of course later phases may override the mime type set in this 
phase.

When overriding the default type_checker handler, which is
usually the mod_mime handler, don't forget to set the
response handler:

  $r->handler('perl-script'); # or $r->handler('modperl');
  $r->set_handlers(PerlResponseHandler => \&handler);

It's the easiest to leave this stage alone and do any desired
settings in the fixups phase.

7.3.10��PerlFixupHandler

The fixups phase is happening just before the content
handling phase.

It gives the last chance to do things before the response is 
generated.

For example in this phase mod_env populates the
environment with variables configured with SetEnv and 
PassEnv directives.

This fixup handler tells Apache at run time which handler and
callback should be used to process the request based on the
file extension of the request's URI.

  package MyApache2::FileExtDispatch;                                #
  
  use strict;
  use warnings;
  
  use Apache2::RequestIO ();
  use Apache2::RequestRec ();
  use Apache2::RequestUtil ();
  
  use Apache2::Const -compile => 'OK';
  
  use constant HANDLER  => 0;
  use constant CALLBACK => 1;
  
  my %exts = (
      cgi => ['perl-script',     \&cgi_handler],
      pl  => ['modperl',         \&pl_handler ],
      tt  => ['perl-script',     \&tt_handler ],
      txt => ['default-handler', undef        ],
  );

  sub handler {                                                     #
      my $r = shift;
  
      my($ext) = $r->uri =~ /\.(\w+)$/;
      $ext = 'txt' unless defined $ext and exists $exts{$ext};
  
      $r->handler($exts{$ext}->[HANDLER]);
  
      if (defined $exts{$ext}->[CALLBACK]) {
          $r->set_handlers(PerlResponseHandler => $exts{$ext}->[CALLBACK]);
      }
  
      return Apache2::Const::OK;
  }

  sub cgi_handler { content_handler($_[0], 'cgi') }
  sub pl_handler  { content_handler($_[0], 'pl')  }
  sub tt_handler  { content_handler($_[0], 'tt')  }
  
  sub content_handler {                                             #
      my($r, $type) = @_;
  
      $r->content_type('text/plain');
      $r->print("A handler of type '$type' was called");
  
      return Apache2::Const::OK;
  }
  
  1;

Here is how this handler is configured:

  Alias /dispatch/ /home/httpd/httpd-2.0/htdocs/
  <Location /dispatch/>
      PerlFixupHandler MyApache2::FileExtDispatch
  </Location>

Notice that there is no need to specify anything, but the fixup 
handler.

It applies the rest of the settings dynamically at run-time.

7.3.11��PerlResponseHandler

The handler (response) phase is used for generating the 
response.

This is arguably the most important phase and most of the
existing Apache modules do most of their work at this phase.

This is the only phase that requires two directives under 
mod_perl:

  <Location /perl>
     SetHandler perl-script
     PerlResponseHandler MyApache2::WorldDomination
  </Location>

SetHandler  set to perl-script  or modperl  tells Apache
that mod_perl is going to handle the response generation.

PerlResponseHandler  tells mod_perl which callback is
going to do the job.

This handler prints itself:

  package MyApache2::Deparse;
  
  use strict;
  use warnings;
  
  use Apache2::RequestRec ();
  use Apache2::RequestIO ();
  use B::Deparse ();
  
  use Apache2::Const -compile => 'OK';
  
  sub handler {
      my $r = shift;
  
      $r->content_type('text/plain');
      $r->print('sub handler ', B::Deparse->new->coderef2text(\&handler));
  
      return Apache2::Const::OK;
  }
  1;

To enable this handler add to httpd.conf:

  <Location /deparse>
      SetHandler modperl
      PerlResponseHandler MyApache2::Deparse
  </Location>

Now when the server is restarted and we issue a request to 
http://localhost/deparse we get the following response:

  sub handler {
      package MyApache2::Deparse;
      use warnings;
      use strict 'refs';
      my $r = shift @_;
      $r->content_type('text/plain');
      $r->print('sub handler ', 'B::Deparse'->new->coderef2text(\&handler));
      return 0;
  }

If you compare it to the source code, it's pretty much the
same code. B::Deparse  is fun to play with!

7.3.12��PerlLogHandler

The log_transaction phase is always executed.

Even if the previous phases were aborted.

At this phase log handlers usually log various information
about the request and the response.

The following handler logs the request data into user-specific 
files.

We assume that all URIs include the username in the form of: 
/~username/

  package MyApache2::LogPerUser;                                     #
  
  use strict;
  use warnings;
  
  use Apache2::RequestRec ();
  use Apache2::Connection ();
  
  use Fcntl qw(:flock);
  use File::Spec::Functions qw(catfile);
  
  use Apache2::Const -compile => qw(OK DECLINED);

  sub handler {                                                     #
      my $r = shift;
  
      my($username) = $r->uri =~ m|^/~([^/]+)|;
      return Apache2::Const::DECLINED unless defined $username;
  
      my $entry = sprintf qq(%s [%s] "%s" %d %d\n),
          $r->connection->remote_ip, scalar(localtime),
          $r->uri, $r->status, $r->bytes_sent;
  
      my $log_path = catfile Apache2::ServerUtil::server_root,
          "logs", "$username.log";
      open my $fh, ">>$log_path" or die "can't open $log_path: $!";
      flock $fh, LOCK_EX;
      print $fh $entry;
      close $fh;
  
      return Apache2::Const::OK;
  }
  1;

Configuration:

  <LocationMatch "^/~">
      SetHandler perl-script
      PerlResponseHandler ModPerl::Registry
      PerlLogHandler MyApache2::LogPerUser
      Options +ExecCGI
  </LocationMatch>

After restarting the server and issuing requests to the
following URIs:

  http://localhost/~stas/test.pl
  http://localhost/~eric/test.pl
  http://localhost/~stas/date.pl

The MyApache2::LogPerUser  handler will append to 
logs/stas.log:

  127.0.0.1 [Sat Aug 31 01:50:38 2002] "/~stas/test.pl" 200 8
  127.0.0.1 [Sat Aug 31 01:50:40 2002] "/~stas/date.pl" 200 44

and to logs/eric.log:

  127.0.0.1 [Sat Aug 31 01:50:39 2002] "/~eric/test.pl" 200 8

7.3.13��PerlCleanupHandler

There is no cleanup Apache phase,

It's a mod_perl convenience invention

It runs immediately after the request has been served

and before the request object is destroyed.

Used to run cleanup code

or to do some time consuming work after the client has gone

Usage:

a cleanup()  callback accepts $r  as its only argument.

  PerlCleanupHandler MyApache2::Cleanup

  or:

  $r->push_handlers(PerlCleanupHandler => \&cleanup);

a cleanup()  callback accepts an optional arbitrary 
argument

  $r->pool->cleanup_register(\&cleanup, $arg);

Example:  delete a temporary file in the cleanup handler

  package MyApache2::Cleanup1;                                       #
  
  use strict;
  use warnings FATAL => 'all';
  
  use File::Spec::Functions qw(catfile);
  
  use Apache2::RequestRec ();
  use Apache2::RequestIO ();
  use Apache2::RequestUtil ();
  
  use Apache2::Const -compile => qw(OK DECLINED);
  use APR::Const    -compile => 'SUCCESS';
  
  my $file = catfile "/tmp", "data";

  sub handler {                                                     #
      my $r = shift;
  
      $r->content_type('text/plain');
  
      local @ENV{qw(PATH BASH_ENV)};
      qx(/bin/ls -l > $file);
  
      my $status = $r->sendfile($file);
      die "sendfile has failed" unless $status == APR::Const::SUCCESS;
  
      $r->push_handlers(PerlCleanupHandler => \&cleanup);
  
      return Apache2::Const::OK;
  }
  
  sub cleanup {
      my $r = shift;
  
      die "Can't find file: $file" unless -e $file;
      unlink $file or die "failed to unlink $file";
  
      return Apache2::Const::OK;
  }
  1;

Next we add the following configuration:

  <Location /cleanup1>
      SetHandler modperl
      PerlResponseHandler MyApache2::Cleanup1
  </Location>

Now when requesting /cleanup1

the contents of the current directory will be printed

and once the request is over the temporary file is deleted.

The problem in Multi-processes/threads env

MyApache2::Cleanup1  is problematic for it uses the same 
filename

Better use a unique filename ($$ + 
APR::OS::current_thread_id())

Even better, use a non-predictable name: rand, APR::UUID

But how do we pass the filename to the cleanup handler?

MyApache2::Cleanup2  solves the problem with:

  $r->pool->cleanup_register(\&cleanup, $file);

  package MyApache2::Cleanup2;                                       #
  
  use strict;
  use warnings FATAL => 'all';
  
  use File::Spec::Functions qw(catfile);
  
  use Apache2::RequestRec ();
  use Apache2::RequestIO ();
  use Apache2::RequestUtil ();
  use APR::UUID ();
  use APR::Pool ();
  
  use Apache2::Const -compile => qw(OK DECLINED);
  use APR::Const    -compile => 'SUCCESS';
  
  my $file_base = catfile "/tmp", "data-";

  sub handler {                                                     #
      my $r = shift;
  
      $r->content_type('text/plain');
      my $file = $file_base . APR::UUID->new->format;
  
      local @ENV{qw(PATH BASH_ENV)};
      qx(/bin/ls -l > $file);
  
      my $status = $r->sendfile($file);
      die "sendfile has failed" unless $status == APR::Const::SUCCESS;
  
      $r->pool->cleanup_register(\&cleanup, $file);
  
      return Apache2::Const::OK;
  }
  sub cleanup {
      my $file = shift;
  
      die "Can't find file: $file" unless -e $file;
      unlink $file or die "failed to unlink $file";
      return Apache2::Const::OK;
  }
  1;

Stas BekmanHTTP Handlers Slide 315



Similarly to the first handler, we add the configuration:

  <Location /cleanup2>
      SetHandler modperl
      PerlResponseHandler MyApache2::Cleanup2
  </Location>

8�� mod_perl 1.0 to
mod_perl 2.0 Migration 

8.1��About
Configuration Directives Changes

mod_perl API Changes

mod_perl Modules Changes

8.2��Migrating from mod_perl
1.0 to mod_perl 2.0

Some configuration directives were renamed or removed

Several APIs have changed, renamed, removed, or moved to
new packages.

...

But there is a back-compatibility layer!

8.3��The Shortest Migration
Path from 1.0
  use Apache2::compat;

Certain Configuration directives and APIs have changed

See http://perl.apache.org/docs/2.0/user/compat/compat.html

8.4��Migrating Configuration 
Files

Several configuration directives are deprecated in 2.0

but still available for backwards compatibility

Otherwise, consider using the directives that have replaced 
them.

8.4.1��PerlHandler
  PerlHandler => PerlResponseHandler

8.4.2��PerlSendHeader
  PerlSendHeader On  => PerlOptions +ParseHeaders
  PerlSendHeader Off => PerlOptions -ParseHeaders

8.4.3��PerlSetupEnv
  PerlSetupEnv On  => PerlOptions +SetupEnv
  PerlSetupEnv Off => PerlOptions -SetupEnv

8.4.4��PerlTaintCheck
The taint mode now can be turned on with:

  PerlSwitches -T

It's disabled by default.

You cannot disable it once it's enabled.

8.4.5��PerlWarn
Warnings now can be enabled globally with:

  PerlSwitches -w

Stas Bekmanmod_perl 1.0 to mod_perl 2.0 Migration Slide 327



8.4.6��PerlFreshRestart
PerlFreshRestart  is a mod_perl 1.0 legacy

In mod_perl 2.0 a full tear-down and startup of interpreters is
done on restart.

To reuse the same httpd.conf for 1.0 and 2.0:

  <IfDefine !MODPERL2>
      PerlFreshRestart
  </IfDefine>

8.5��Code Porting
Certain APIs couldn't be preserved in 2.0, hence:

  use Apache2::compat;

which is slower, because it's implemented in Perl.

it's also incomplete, since some APIs can't be backported

Consider getting rid of using the compat layer

8.5.1��Finding Which Modules
Need To Be Loaded
The Poison:

  $r->content_type('text/plain');
  $r->print("Hello cruel world!");

the execution fails with:

  can't find method 'content_type' ...

One has to load modules containing mod_perl methods
before they can be used

The Remedy :

  % perl -MModPerl::MethodLookup -e print_method content_type
  to use method 'content_type' add:
           use Apache2::RequestRec ();

Alias :

  # c-style shell
  % alias lookup "perl -MModPerl::MethodLookup -e print_method"
  
  # bash-style shell
  % alias lookup="perl -MModPerl::MethodLookup -e print_method"
  
  % lookup content_type
  to use method 'content_type' add:
           use Apache2::RequestRec ();

8.6��ModPerl::Registry Family
s/Apache::Registry/ModPerl::Registry/

  Alias /perl/ /home/httpd/perl/
  <Location /perl>
      SetHandler perl-script
      PerlResponseHandler ModPerl::Registry
      Options +ExecCGI
      PerlOptions +ParseHeaders
  </Location>

MPM doesn't matter, mp2 should work the same everywhere

Issues with chdir(), which affects all threads: use 
ModPerl::RegistryPrefork  or 
ModPerl::PerlRunPrefork

Cook your own registry with ModPerl::RegistryCooker

is-a: Subclass one of the classes

has-a: Build your own using aliasing

8.7��Method Handlers
$$ doesn't work since some callbacks accepts more than 2 
args

  package Bird;
  @ISA = qw(Eagle);
  
  sub handler : method {
      my($class, $r) = @_;
      ...;
  }

See the attributes manpage.

8.8��Apache::StatINC 
Replacement

Replace:

  PerlInitHandler Apache::StatINC

with:

  PerlInitHandler Apache2::Reload

Apache2::Reload  provides an extra functionality

See the module's manpage.

9�� That's all folks! 

9.1��Thanks
Thanks to TicketMaster for sponsoring my work on mod_perl

9.2��References
mod_perl 2.0 information can be found at:

  http://perl.apache.org/docs/2.0/

Further Questions?

Grab me at the corridor and demand answers

Ask at modperl@perl.apache.org

9.3��A Shameless Plug

Now available under Creative Commons license: 

download

modify

distribute

sell

etc.

Get it from http://modperlbook.org/

